

5/12

Tracé de rectangles vides et pleins

Nous vous proposons les RSX !BOX et !PBOX qui affichent respectivement des rectangles vides et pleins sur l'écran.

RSX !BOX

Comment utiliser la RSX

Le format de la RSX !BOX est le suivant :

!BOX, X1, Y1, X2, Y2, C

où X1 et Y1 sont les coordonnées du coin supérieur gauche du rectangle, X2 et Y2 sont les coordonnées du coin inférieur droit du rectangle, et C est la couleur de tracé. Les valeurs possibles pour C dépendent du mode d'affichage :


MODE 0 : 0 à 15,

MODE 1 : 0 à 3,

MODE 2 : 0 ou 1.

La RSX en détail

La logique de la RSX apparaît dans l'ordinogramme suivant :



La routine de traitement se trouve à l'étiquette **BOX**. Les données qui lui sont passées sont stockées dans les variables **COUL** (couleur), **X1**, **Y1**, **X2** et **Y2** (coordonnées extrémales) entre les lignes 56 et 69.

La couleur de tracé est initialisée à l'aide de la macro **GRA SET PEN** du FIRMWARE (lignes 76 et 77).

Le tracé du rectangle vide se fait en quatre étapes.

- Etape 1 :
 - origine du tracé en X1, Y1 (lignes 83 à 85) ;
 - tracé d'une ligne entre X1, Y1 et X2, Y1 (lignes 86 à 88).
- Etape 2 :
 - origine du tracé en X1, Y1 (lignes 90 à 92) ;
 - tracé d'une ligne entre X1, Y1 et X1, Y2 (lignes 93 à 95).
- Etape 3 :
 - origine du tracé en X2, Y1 (lignes 97 à 99) ;
 - tracé d'une ligne entre X2, Y1 et X2, Y2 (lignes 100 à 102).
- Etape 4 :
 - origine du tracé en X1, Y2 (lignes 104 à 106) ;
 - tracé d'une ligne entre X1, Y2 et X2, Y2 (lignes 107 à 109).

Le positionnement de l'origine de chaque tracé se fait avec la macro **GRA MOV ABS** du FIRMWARE, et le tracé des lignes avec la macro **GRA LINE ABS** du FIRMWARE.

Le listing de la RSX est le suivant :

```

1          ORG  9000H
2          LOAD 9000H
3          ;-----
4          ; RSX BOX
5          ; Format : ;BOX,X1,Y1,X2,Y2,C
6          ; Entree : X1=Abs coin sup gauche
7          ;          Y1=Ord coin sup gauche
8          ;          X2=Abs coin inf droit
9          ;          Y2=Ord coin inf droit
10         ;          C=Couleur
11         ; Sortie : Affichage du rectangle
12         ;-----
13         ;
14         ;
15         ;-----
16         ; Declaration des constantes
17         ; et des variables du programme
18         ;-----
19         ;
20         LOGEXT:   EQU  0BCD1H           ;KL LOG EXT
21         SETPEN:   EQU  0BBDEH           ;GRA SET PEN
22         MOVABS:   EQU  0BBC0H           ;GRA MOVE ABS
23         LINEABS:  EQU  0BBF6H           ;GRA LINE ABS
24         BUF:      DS    4               ;ZONE RAM POUR LOG EXT
25 9004 0990        PTRTAB:  DW  TABLE     ;Pointeur TABLE
26 9006 C32090      JP  BOX               ;Affichage du rectangle
27 9009 424F        TABLE:  DB  "BO"

```

```

28 900B D8          DB  "X"+80H
29 900C 00          DB  0          ;Fin de table
30                X1:    DS  2          ;Abscisse coin sup gauche
31                Y1:    DS  2          ;Ordonnee coin sup gauche
32                X2:    DS  2          ;Abscisse coin inf droit
33                Y2:    DS  2          ;Ordonnee coin inf droit
34                COUL:   DS  1          ;Couleur de trace
35                ;
36                ;-----
37                ; Definition de la RSX
38                ;-----
39                ;
40                DEFRSX: EQU  $          ;Point d'entree
41 9016 010490      LD  BC,PTRTAB      ;Ptr table definition
42 9019 210090      LD  HL,BUF        ;Buffer pour LOG EXT
43 901C CDD1BC      CALL LOGEXT       ;Definition de la RSX
44 901F C9          RET
45                ;
46                ;-----
47                ; Traitement de BOX
48                ;-----
49                ;
50                BOX:    EQU  $          ;Point d'entree
51                ;
52                ;-----
53                ; Lecture des donnees passees
54                ;-----

```

```

55          ;
56 9020 DD7E00      LD  A,(IX+0)
57 9023 321590      LD  (COUL),A          ;Couleur de trace
58 9026 DD6603      LD  H,(IX+3)
59 9029 DD6E02      LD  L,(IX+2)
60 902C 221390      LD  (Y2),HL          ;Ord coin inf droit
61 902F DD6605      LD  H,(IX+5)
62 9032 DD6E04      LD  L,(IX+4)
63 9035 221190      LD  (X2),HL          ;Abs coin inf droit
64 9038 DD6607      LD  H,(IX+7)
65 903B DD6E06      LD  L,(IX+6)
66 903E 220F90      LD  (Y1),HL          ;Ord coin sup gauche
67 9041 DD6609      LD  H,(IX+9)
68 9044 DD6E08      LD  L,(IX+8)
69 9047 220D90      LD  (X1),HL          ;Abs coin sup gauche
70          ;
71          ;-----
72          ; Initialisation de la couleur
73          ; de trace du rectangle
74          ;-----
75          ;
76 904A 3A1590      LD  A,(COUL)
77 904D CDDEBB      CALL SETPEN          ;Couleur de trace
78          ;
79          ;-----
80          ; Trace du rectangle
81          ;-----

```

```
82          ;
83 9050 ED5B0D90      LD  DE, (X1)
84 9054 2A0F90       LD  HL, (Y1)
85 9057 CDC0BB       CALL MOVABS           ;Coin sup gauche
86 905A ED5B1190     LD  DE, (X2)
87 905E 2A0F90       LD  HL, (Y1)
88 9061 CDF6BB       CALL LINEABS         ;Ligne du haut
89          ;
90 9064 ED5B0D90     LD  DE, (X1)
91 9068 2A0F90       LD  HL, (Y1)
92 906B CDC0BB       CALL MOVABS           ;Coin sup gauche
93 906E ED5B0D90     LD  DE, (X1)
94 9072 2A1390       LD  HL, (Y2)
95 9075 CDF6BB       CALL LINEABS         ;Ligne de gauche
96          ;
97 9078 ED5B1190     LD  DE, (X2)
98 907C 2A0F90       LD  HL, (Y1)
99 907F CDC0BB       CALL MOVABS           ;Coin sup droit
100 9082 ED5B1190    LD  DE, (X2)
101 9086 2A1390       LD  HL, (Y2)
102 9089 CDF6BB       CALL LINEABS         ;Ligne de droite
103          ;
104 908C ED5B0D90     LD  DE, (X1)
105 9090 2A1390       LD  HL, (Y2)
106 9093 CDC0BB       CALL MOVABS           ;Coin inf gauche
107 9096 ED5B1190     LD  DE, (X2)
108 909A 2A1390       LD  HL, (Y2)
109 909D CDF6BB       CALL LINEABS         ;Ligne du bas
```

```

110          ;
111 90A0 C9          RET
112          END

BUF          9000 BOX          9020 COUL          9015 DEFERSX          9016
LOGEXT      BCD1 LINEABS      BBF6 MOVABS          BBC0 PTRTAB          9004
SETPEN      BBDE TABLE      9009 X1          900D X2          9011
Y1          900F Y2          9013

```

Comme toujours, voici le chargeur Basic correspondant :

```

1000 REM -----
1010 REM Chargeur BASIC de la RSX de trace de rectangles vides
1020 REM -----
1030 REM
1040 FOR i=&9000 TO &90A0
1050   READ a$
1060   a$="&" + a$
1070   POKE i,VAL(a$)
1080 NEXT i
1090 CALL &9016
1100 END
1110 REM - - - - -
1120 REM Donnees du programme de trace
1130 REM - - - - -
1140 REM
1150 REM BOX
1160 DATA 0,0,0,0,9,90,C3,20,90,42,4F,D8,0,0,0,0
1170 DATA 0,0,0,0,0,0,1,4,90,21,0,90,CD,D1,BC,C9
1180 DATA DD,7E,0,32,15,90,DD,66,3,DD,6E,2,22,13,90,DD
1190 DATA 66,5,DD,6E,4,22,11,90,DD,66,7,DD,6E,6,22,F
1200 DATA 90,DD,66,9,DD,6E,8,22,D,90,3A,15,90,CD,DE,BB
1210 DATA ED,5B,D,90,2A,F,90,CD,C0,BB,ED,5B,11,90,2A,F
1220 DATA 90,CD,F6,BB,ED,5B,D,90,2A,F,90,CD,C0,BB,ED,5B
1230 DATA D,90,2A,13,90,CD,F6,BB,ED,5B,11,90,2A,F,90,CD
1240 DATA C0,BB,ED,5B,11,90,2A,13,90,CD,F6,BB,ED,5B,D,90
1250 DATA 2A,13,90,CD,C0,BB,ED,5B,11,90,2A,13,90,CD,F6,BB
1260 DATA C9,0,0,0,0,0,0,0,0,0,0,0,0,0,0

```

et les données de checksum afférentes :

```
78 6D 6D 4E 3A 1F 55 6E 9C 51 C9
```

Le petit programme de démonstration suivant montre combien la RSX !BOX est facile à manipuler :

```
100 REM -----
110 REM Programme de demonstration
120 REM -----
130 REM
140 MODE 1
150 FOR I=1 TO 50
160   X1=INT(RND(1)*500)
170   Y1=INT(RND(1)*300)
180   X2=INT(RND(1)*600)
190   Y2=INT(RND(1)*400)
200   C=INT(RND(1)*3)+1
210   !BOX,X1,Y1,X2,Y2,C
220 NEXT I
```

Avant de pouvoir utiliser !BOX, il faut bien entendu l'avoir installée, par exemple avec le chargeur Basic précédent.

RSX ; PBOX

Comment utiliser la RSX


Le format de la RSX !PBOX est le même que celui de la RSX !BOX :

```
!BOX, X1, Y1, X2, Y2, C
```

où X1 et Y1 sont les coordonnées du coin supérieur gauche du rectangle, X2 et Y2 sont les coordonnées du coin inférieur droit du rectangle, et C est la couleur de tracé.

La RSX en détail

La logique de la RSX apparaît dans l'ordinogramme suivant :



La routine de traitement se trouve à l'étiquette **PBOX**. Les données qui lui sont passées sont stockées dans les variables **COUL** (couleur), **X1**, **Y1**, **X2** et **Y2** (coordonnées extrémales) entre les lignes 57 et 70.

La couleur de tracé est initialisée à l'aide de la macro **GRA SET PEN** du **FIRMWARE** (lignes 77 et 78).

Avant de commencer le tracé du rectangle, le programme calcule :

- La longueur du rectangle (lignes 84 à 88). A ce propos, remarquez l'instruction **XOR A** ligne 86 avant d'effectuer la soustraction. Cette instruction met à zéro l'éventuel indicateur de retenue dans le registre des indicateurs. Si cet indicateur était à un, la soustraction ne donnerait pas le résultat attendu.
- La hauteur du rectangle (lignes 90 à 94). La hauteur, c'est-à-dire le nombre de lignes élémentaires du rectangle, est stockée dans le registre B. Ce registre sera utilisé dans la boucle principale du programme pour décompter le nombre de lignes élémentaires tracées.

La boucle de tracé se trouve entre les lignes 98 et 110. La macro **GRA MOVE ABS** positionne l'origine du tracé, et la macro **GRA LINE REL** trace une ligne horizontale de longueur **OFFX** à partir de l'origine du tracé. Une instruction **DJNZ** décrémente le registre B et redonne le contrôle à l'étiquette **BIS** tant que B n'est pas nul, c'est-à-dire tant que tout le rectangle n'est pas tracé.

Le listing de la RSX est le suivant :

```

1          ORG  9000H
2          LOAD 9000H
3          ;-----
4          ; RSX FBOX
5          ; Format : !BOX,X1,Y1,X2,Y2,C
6          ; Entree : X1=Abs coin sup gauche
7          ;          Y1=Ord coin sup gauche
8          ;          X2=Abs coin inf droit
9          ;          Y2=Ord coin inf droit
10         ;          C=Couleur
11         ; Sortie : Affichage du rectangle
12         ;-----
13         ;
14         ;
15         ;-----
16         ; Declaration des constantes
17         ; et des variables du programme
18         ;-----
19         ;
20         LOGEXT:    EQU  0BCD1H          ;KL LOG EXT
21         SETPEN:   EQU  0BBDEH          ;GRA SET PEN
22         MOVABS:   EQU  0BBC0H          ;GRA MOVE ABS
23         LINEREL:  EQU  0BBF9H          ;GRA LINE REL
24         BUF:      DS    4              ;ZONE RAM POUR LOG EXT
25 9004 0990        PTRTAB:    DW    TABLE          ;Pointeur TABLE
26 9006 C32390     JP    FBOX          ;Affichage du rectangle
27 9009 50424F     TABLE:    DB    "FBO"

```

```

28 900C DB          DB  "X"+80H
29 900D 00          DB  0           ;Fin de table
30          OFFX:    DS  2           ;Offset en X
31          X1:      DS  2           ;Abscisse coin sup gauche
32          Y1:      DS  2           ;Ordonnee coin sup gauche
33          X2:      DS  2           ;Abscisse coin inf droit
34          Y2:      DS  2           ;Ordonnee coin inf droit
35          COUL:    DS  1           ;Couleur de trace
36          ;
37          ;-----
38          ; Definition de la RSX
39          ;-----
40          ;
41          DEFRSX:  EQU  $           ;Point d'entree
42 9019 010490      LD  BC,FTRTAB     ;Ptr table definition
43 901C 210090      LD  HL,BUF        ;Buffer pour LOG EXT
44 901F CDD1BC      CALL LOGEXT      ;Definition de la RSX
45 9022 C9          RET
46          ;
47          ;-----
48          ; Traitement de BOX
49          ;-----
50          ;
51          FBOX:    EQU  $           ;Point d'entree
52          ;
53          ;-----
54          ; Lecture des donnees passees

```

```

55      ; - - - - -
56      ;
57 9023 DD7E00      LD  A, (IX+0)
58 9026 321890      LD  (COUL), A      ;Couleur de trace
59 9029 DD6603      LD  H, (IX+3)
60 902C DD6E02      LD  L, (IX+2)
61 902F 221690      LD  (Y2), HL      ;Ord coin inf droit
62 9032 DD6605      LD  H, (IX+5)
63 9035 DD6E04      LD  L, (IX+4)
64 9038 221490      LD  (X2), HL      ;Abs coin inf droit
65 903B DD6607      LD  H, (IX+7)
66 903E DD6E06      LD  L, (IX+6)
67 9041 221290      LD  (Y1), HL      ;Ord coin sup gauche
68 9044 DD6609      LD  H, (IX+9)
69 9047 DD6E08      LD  L, (IX+8)
70 904A 221090      LD  (X1), HL      ;Abs coin sup gauche
71      ;
72      ; - - - - -
73      ; Initialisation de la couleur
74      ; du rectangle
75      ; - - - - -
76      ;
77 904D 3A1890      LD  A, (COUL)
78 9050 CDDEBB      CALL SETPEN      ;Couleur rectangle
79      ;
80      ; - - - - -
81      ; Trace du rectangle plein

```

```

82          ; - - - - -
83          ;
84 9053 2A1490          LD  HL, (X2)
85 9056 ED5B1090       LD  DE, (X1)
86 905A AF             XOR  A
87 905B ED52           SBC  HL, DE          ; X2-X1
88 905D 220E90        LD  (OFFX), HL      ; Offset en X
89          ;
90 9060 2A1690       LD  HL, (Y2)
91 9063 ED5B1290     LD  DE, (Y1)
92 9067 AF           XOR  A
93 9068 ED52         SBC  HL, DE
94 906A 45           LD  B, L          ; Y2-Y1
95          ;
96 906B ED5B1090     LD  DE, (X1)
97 906F 2A1290      LD  HL, (Y1)
98          BIS:     EQU  $
99 9072 C5           PUSH BC
100 9073 D5          PUSH DE
101 9074 E5          PUSH HL
102 9075 CDC0BB      CALL MOVABS          ; Coin sup gauche
103 9078 ED5B0E90   LD  DE, (OFFX)
104 907C 210000     LD  HL, 0
105 907F CDF9BB     CALL LINEREL        ; Une ligne
106 9082 E1         POP  HL
107 9083 D1         POP  DE
108 9084 C1         POP  BC

```

```

109 9085 23          INC HL
110 9086 10EA       DJNZ BIS          ; Boucle
111                ;
112 9088 C9         RET
113                END

BUF          9000 BIS          9072 COUL          9018 DEFRSX          9019
LOGEXT      BCD1 LINEREL      BBF9 MOVABS          BBC0 OFFX          900E
PTRTAB      9004 PBOX          9023 SETPEN          BBDE TABLE          9009
X1          9010 X2          9014 Y1          9012 Y2          9016

```

Comme toujours, voici le chargeur Basic correspondant :

```

1000 REM -----
1010 REM Chargeur BASIC de la RSX de trace de rectangles pleins
1020 REM -----
1030 REM
1040 FOR i=&9000 TO &9088
1050   READ a#
1060   a#="%" +a#
1070   POKE i,VAL(a#)
1080 NEXT i
1090 CALL &9019
1100 END
1110 REM - - - - -
1120 REM Donnees du programme de trace
1130 REM - - - - -
1140 REM
1150 DATA 0,0,0,0,9,90,C3,23,90,50,42,4F,DB,0,0,0
1160 DATA 0,0,0,0,0,0,0,0,1,4,90,21,0,90,CD
1170 DATA D1,BC,C9,DD,7E,0,32,18,90,DD,66,3,DD,6E,2,22
1180 DATA 16,90,DD,66,5,DD,6E,4,22,14,90,DD,66,7,DD,6E
1190 DATA 6,22,12,90,DD,66,9,DD,6E,8,22,10,90,3A,18,90
1200 DATA CD,DE,BB,2A,14,90,ED,5B,10,90,AF,ED,52,22,E,90
1210 DATA 2A,16,90,ED,5B,12,90,AF,ED,52,45,ED,5B,10,90,2A
1220 DATA 12,90,CS,D5,ES,CD,C0,BB,ED,5B,E,90,21,0,0,CD
1230 DATA F9,BB,E1,D1,C1,23,10,EA,C9,0,0,0,0,0,0,0

```

et les données de checksum qui permettent de vérifier l'exactitude des données entrées :

```
CB 15 47 9E 12 D1 6 45 13
```

Le petit programme de démonstration suivant montre comment utiliser la RSX !PBOX :

```
100 REM -----
110 REM Programme de demonstration
120 REM -----
130 REM
140 MODE 1
150 FOR I=1 TO 50
160   X1=INT(RND(1)*500)
170   Y1=INT(RND(1)*300)
180   X2=INT(RND(1)*600)
190   Y2=INT(RND(1)*400)
200   C=INT(RND(1)*3)+1
210   !PBOX,X1,Y1,X2,Y2,C
220 NEXT I
```

